Drgania własne belki swobodnie podpartej

ORIGIN $:= 1$

$E := 10\ \mathbf{GPa}$ $\quad b := 10\ \mathbf{cm}$ $\quad h := 10\ \mathbf{cm}$ $\quad L := 10\ \mathbf{m}$ $\quad \rho := 600\ \dfrac{\mathbf{kg}}{\mathbf{m}^3}$

$A := b \cdot h$ $\quad J := \dfrac{A \cdot h^2}{12}$ $\quad \mu := \rho \cdot A$ $\quad c := \sqrt{\dfrac{E \cdot J}{\mu}}$

$A = 100\ cm^2$ $\quad J = 833.333\ cm^4$ $\quad \mu = 6\ \dfrac{\mathbf{kg}}{\mathbf{m}}$ $\quad c = 117.851\ \dfrac{\mathbf{m}^2}{\mathbf{s}}$ $\quad EJ := E \cdot J = 83.333333\ kN \cdot m^2$

Warunki brzegowe: $\quad y(0) = 0 \quad y(L) = 0$

$M(0) = 0 \quad M(L) = 0$

Równanie różniczkowe opisujące swobodne drgania pręta prostego przy zastosowaniu modelu Bernoulliego (bez udziału sił poprzeczych i bezwładności obrotowej).

$$\frac{M(x)}{EJ} = \frac{d^2}{dx^2} y(x)$$

$$EJ \cdot \frac{d^4}{dx^4} w(x,t) + \mu \cdot \frac{d^2}{dt^2} w(x,t) = 0 \qquad (1)$$

Przyjmując rozwiązanie w postaci $\quad w(x,t) = y(x) \cdot sin(\omega \cdot t)$ $\quad$ *otrzymamy:*

$$\left[\frac{d^4}{dx^4} y(x) - \frac{\mu \cdot \omega^2}{EJ} \cdot y(x)\right] \cdot sin(\omega \cdot t) = 0 \quad \text{--------->} \quad \frac{d^4}{dx^4} y(x) - \frac{\varphi^4}{L^4} \cdot y(x) = 0 \qquad \varphi = L \cdot \sqrt{\frac{\omega}{c}}$$

Równanie przestępne, wynikające z warunków brzegowych, dające częstości drgań własnych: $\quad sin(\varphi) = 0$



$n := 6$

$i := 1..n \qquad \varphi_i := i \cdot \pi \qquad \omega_A := \dfrac{c}{L^2} \cdot \varphi^2$

Postacie drgań własnych

$W(n,x) := \sin\left(\varphi_n \cdot x\right)$

$$\varphi = \begin{bmatrix} 3.141593 \\ 6.283185 \\ 9.424778 \\ 12.566371 \\ 15.707963 \\ 18.849556 \end{bmatrix} \qquad \omega_A = \begin{bmatrix} 11.63144 \\ 46.525761 \\ 104.682963 \\ 186.103045 \\ 290.786008 \\ 418.731852 \end{bmatrix} \dfrac{\mathbf{rad}}{\mathbf{s}}$$



Ortogonalność postaci drgań

$i := 1..n \qquad j := 1..n$

$$C_{i,j} := \int_0^1 W(i,x) \cdot W(j,x)\, \mathrm{d}x \qquad C = \begin{bmatrix} 0.5 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0.5 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0.5 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0.5 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0.5 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0.5 \end{bmatrix}$$

Drgania własne belki obliczone metodą elementów skończonych



$l := 1\ m$ <--- jednostka długości lub długość porównawcza

$N := 5$ $\quad Le := \frac{L}{N}$ $\quad \kappa := \frac{EJ}{Le^2} = 20.833333\ kN$ $\quad \lambda e := \frac{Le}{l} = 2$ $\quad Lss := 2$ $\quad Lw := N+1$ - Liczba węzłów

$Lr := Lss \cdot Lw$ - Liczba równań

Funkcja DBM - Dodaj Blok Macierzy

```
DBM(A,B,w,k) := | for i ∈ 1..rows(B)
                |   for j ∈ 1..cols(B)
                |     A[w+i-1, k+j-1] ← B[i,j] + A[w+i-1, k+j-1]
                | A
```

Agregacja macierzy globalnych belki

$$Ae = \begin{bmatrix} Aa & Ab \\ Ab^T & Ac \end{bmatrix}$$

```
Agrg_B(N,Aa,Ab,Ac) := | Lss ← 2
                      | Lr ← Lss (N+1)
                      | A[Lr,Lr] ← 0
                      | for e ∈ 1..N
                      |   n ← Lss e-1
                      |   k ← n+2
                      |   A ← DBM(A,Aa,n,n)
                      |   A ← DBM(A,Ac,k,k)
                      |   A ← DBM(A,Ab,n,k)
                      |   A ← DBM(A,Ab^T,k,n)
                      | A
```

Macierz sztywności elementów belki

$$K = \frac{EJ}{Le^2} \cdot \begin{bmatrix} \frac{12}{Le} & 6 & \frac{-12}{Le} & 6 \\ 6 & 4 \cdot Le & -6 & 2 \cdot Le \\ \frac{-12}{Le} & -6 & \frac{12}{Le} & -6 \\ 6 & 2 \cdot Le & -6 & 4 \cdot Le \end{bmatrix}$$

$$K = \kappa \cdot \begin{bmatrix} Ka & Kb \\ Kb^{\mathrm{T}} & Kc \end{bmatrix}$$

$\kappa = 20.833333$ ***kN***

Bloki bezwymiarowej macierzy sztywności elementu

$$Ka := \begin{bmatrix} \frac{12}{\lambda e} & 6 \\ 6 & 4 \cdot \lambda e \end{bmatrix} \qquad Kb := \begin{bmatrix} \frac{-12}{\lambda e} & 6 \\ -6 & 2 \cdot \lambda e \end{bmatrix} \qquad Kc := \begin{bmatrix} \frac{12}{\lambda e} & -6 \\ -6 & 4 \cdot \lambda e \end{bmatrix}$$

Agregacja globalnej bezwymiarowej macierzy sztywności belki

$K := Agrg\_B(N, Ka, Kb, Kc)$

$$K = \begin{bmatrix} 6 & 6 & -6 & 6 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 6 & 8 & -6 & 4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -6 & -6 & 12 & 0 & -6 & 6 & 0 & 0 & 0 & 0 & 0 & 0 \\ 6 & 4 & 0 & 16 & -6 & 4 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -6 & -6 & 12 & 0 & -6 & 6 & 0 & 0 & 0 & 0 \\ 0 & 0 & 6 & 4 & 0 & 16 & -6 & 4 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -6 & -6 & 12 & 0 & -6 & 6 & 0 & 0 \\ 0 & 0 & 0 & 0 & 6 & 4 & 0 & 16 & -6 & 4 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -6 & -6 & 12 & 0 & -6 & 6 \\ 0 & 0 & 0 & 0 & 0 & 0 & 6 & 4 & 0 & 16 & -6 & 4 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -6 & -6 & 6 & -6 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 6 & 4 & -6 & 8 \end{bmatrix}$$

Konsekwentna macierz bezwładności elementów belki

$$M=\frac{\mu\cdot Le^2}{420}\cdot\begin{bmatrix}\frac{156}{Le} & 22 & \frac{54}{Le} & -13\\ 22 & 4\cdot Le & 13 & -3\cdot Le\\ \frac{54}{Le} & 13 & \frac{156}{Le} & -22\\ -13 & -3\cdot Le & -22 & 4\cdot Le\end{bmatrix}\qquad M=\eta\cdot\begin{bmatrix}Ma & Mb\\ Mb^{\mathrm{T}} & Mc\end{bmatrix}$$

$$\eta:=\frac{\mu\cdot Le^2}{420}=5.714286\ kg\cdot cm$$

$$\beta_k:=\frac{\eta}{\kappa}=\left(2.742857\cdot 10^{-6}\right)\ \boldsymbol{s}^2$$

Bloki bezwymiarowej macierzy bezwładności

$$Ma:=\begin{bmatrix}\frac{156}{\lambda e} & 22\\ 22 & 4\cdot\lambda e\end{bmatrix}\qquad Mb:=\begin{bmatrix}\frac{54}{\lambda e} & -13\\ 13 & -3\cdot\lambda e\end{bmatrix}\qquad Mc:=\begin{bmatrix}\frac{156}{\lambda e} & -22\\ -22 & 4\cdot\lambda e\end{bmatrix}$$

Agregacja bezwymiarowej konsekwentnej macierzy bezwładności

$$Mk:=Agrg\_B(N,Ma,Mb,Mc)$$

$$Mk=\begin{bmatrix}
78 & 22 & 27 & -13 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0\\
22 & 8 & 13 & -6 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0\\
27 & 13 & 156 & 0 & 27 & -13 & 0 & 0 & 0 & 0 & 0 & 0\\
-13 & -6 & 0 & 16 & 13 & -6 & 0 & 0 & 0 & 0 & 0 & 0\\
0 & 0 & 27 & 13 & 156 & 0 & 27 & -13 & 0 & 0 & 0 & 0\\
0 & 0 & -13 & -6 & 0 & 16 & 13 & -6 & 0 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 27 & 13 & 156 & 0 & 27 & -13 & 0 & 0\\
0 & 0 & 0 & 0 & -13 & -6 & 0 & 16 & 13 & -6 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & 0 & 27 & 13 & 156 & 0 & 27 & -13\\
0 & 0 & 0 & 0 & 0 & 0 & -13 & -6 & 0 & 16 & 13 & -6\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 27 & 13 & 78 & -22\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -13 & -6 & -22 & 8
\end{bmatrix}$$

Diagonalna macierz bezwładności elementów belki

$$M = \frac{\mu \cdot Le^2}{24} \cdot \begin{bmatrix} \frac{12}{Le} & 0 & 0 & 0 \\ 0 & Le & 0 & 0 \\ 0 & 0 & \frac{12}{Le} & 0 \\ 0 & 0 & 0 & Le \end{bmatrix} \qquad M = \eta \cdot \begin{bmatrix} Ma & 0 \\ 0 & Ma \end{bmatrix}$$

$$\eta := \frac{\mu \cdot Le^2}{24} = 1 \ \mathbf{kg \cdot m} \qquad Ma := \begin{bmatrix} \frac{12}{\lambda e} & 0 \\ 0 & \lambda e \end{bmatrix} \qquad M0 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

$$\beta_d := \frac{\eta}{\kappa} = \left(4.8 \cdot 10^{-5}\right) \ \mathbf{s}^2$$

Aby zmniejszyć wpływ bezwładności obrotowej można przyjąć Ma ze współczynnikiem 0<a<1

$$a := 0.1 \qquad Ma := \begin{bmatrix} \frac{12}{\lambda e} & 0 \\ 0 & a \cdot \lambda e \end{bmatrix}$$

Agregacja diagonalnej bezwymiarowej macierzy bezwładności

$$Md := Agrg\_B(N, Ma, M0, Ma)$$

$$Md = \begin{bmatrix}
6 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0.2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 12 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0.4 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 12 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0.4 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 12 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.4 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 12 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.4 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 6 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.2
\end{bmatrix}$$

Diagonalna macierz bezwładności elementów belki, w której pominięto bezwładność obrotową

$$M = \frac{\mu \cdot Le^2}{2} \cdot \begin{bmatrix} \frac{1}{Le} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{1}{Le} & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \qquad M = \eta \cdot \begin{bmatrix} Mb & 0 \\ 0 & Mb \end{bmatrix}$$

$$\eta := \frac{\mu \cdot Le^2}{2} \qquad Mb := \begin{bmatrix} \frac{1}{\lambda e} & 0 \\ 0 & 0 \end{bmatrix}$$

$$\beta_r := \frac{\eta}{\kappa} = \left(5.76 \cdot 10^{-4}\right)\ \mathbf{s}^2$$

Agregacja diagonalnej bezwymiarowej macierzy bezwładności z pominięciem bezwładności obrotowej

$$Mr := Agrg\_B(N, Mb, M0, Mb)$$

$$Mr = \begin{bmatrix}
0.5 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0.5 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0
\end{bmatrix}$$

1. Bezwymiarowa konsekwentna macierz bezwładności $M = Mk$

2. Bezwymiarowa diagonalna macierz bezwładności $M = Md$

3. Bezwymiarowa diagonalna macierz bezwładności bez bezwładności rotacyjnych $M = Mr$

$M := Md$ *- przyjmujemy macierz diagonalną, która ułatwi obliczenie częstości własnych*

Uwzględnienie warunków brzegowych

$i := 1 .. Lr$

$w := 1$ $\quad K_{w,i} := 0 \quad K_{i,w} := 0 \quad K_{w,w} := 1 \quad M_{w,i} := 0 \quad M_{w,w} := 1$

$w := Lr - 1$ $\quad K_{w,i} := 0 \quad K_{i,w} := 0 \quad K_{w,w} := 1 \quad M_{w,i} := 0 \quad M_{w,w} := 1$

$\kappa \cdot \left[K - \omega^2 \ \beta_d \cdot M\right] \ y = 0$ $\quad |M| = 2.123366 \cdot 10$ $\quad M1 := M^{-1}$

Ponieważ diagonalna macierz bezwładności nie jest osobliwa, można zadanie doprowadzić do standardowego zadania na wartości własne. Mnożąc wyrażenie w nawiasie kwadratowym lewostronnie przez odwrotność macierzy bezwładności M1, mamy:

$\left|M1 \cdot K - \omega^2 \cdot \beta_d \cdot I\right| = 0$ $\quad \omega^2 \cdot \beta_d = \sigma$

*Iloczyn M1*K nie jest macierzą symetryczną jak to wymaga standardowy problem wartosci własnych, ale gdy korzystamy z funkcji MathCada "eigenvals" nie musimy symetryzować macierzy M1*K*

$\sigma := \text{eigenvals}(M1 \cdot K)$ $\quad \omega := \sqrt{\frac{\sigma}{\beta_d}}$ $\quad \omega := \text{sort}(\omega)$

$$\omega = \begin{bmatrix} 11.611008 \\ 46.117591 \\ 101.525192 \\ 144.337567 \\ 144.337567 \\ 167.567469 \\ 645.497224 \\ \vdots \end{bmatrix} \frac{\mathbf{rad}}{\mathbf{s}}$$

Analitycznie obliczone częstości własne

$$\omega_A = \begin{bmatrix} 11.63144 \\ 46.525761 \\ 104.682963 \\ 186.103045 \\ 290.786008 \\ 418.731852 \end{bmatrix} \frac{\mathbf{rad}}{\mathbf{s}}$$

Różnice wynikające z różnych metod obliczeniowych

$$i := 1 .. \mathrm{rows}\left(\omega_A\right) \qquad \delta_i := \omega_{A_i} - \omega_i \qquad \delta = \begin{bmatrix} 0.020432 \\ 0.40817 \\ 3.157771 \\ 41.765478 \\ 146.448441 \\ 251.164383 \end{bmatrix} \frac{1}{\mathbf{s}}$$

Jak widać, wyliczone analitycznie częstości są większe niż wyliczone numerycznie z użyciem macierzy diagonalnej z elementami bezwładności obrotowej. Wynika to stąd, że równanie różniczkowe (1) nie uwzględnia bezwładności obrotowej, więc mamy mniejszą "bezwładność analityczną" a stąd wyższe częstości własne.

$N = 5$

$$\delta_1 := \begin{bmatrix} 0.187967 \\ 2.862592 \\ 13.479993 \\ 41.765478 \\ 146.448441 \\ 271.685943 \end{bmatrix} \qquad a = 1$$

Gdy zmniejszymy 1000 razy bezwładność obrotową (biorąc a=0.001 w równaniu na macierz bezwładnosci) otrzymamy dużo mniejsze różnicze w wynikach analitycznych i MES:

$$\delta_{0.001} = \begin{bmatrix} 0.000017 \\ 0.000513 \\ 0.004746 \\ 0.025154 \\ 0.096151 \\ 0.296993 \end{bmatrix}$$

$$\delta_{0.1} := \begin{bmatrix} 0.020432 \\ 0.40817 \\ 3.157771 \\ 41.765478 \\ 146.448441 \\ 251.164383 \end{bmatrix} \qquad a = 0.1$$

$$\delta_{0.01} := \begin{bmatrix} 0.00329 \\ 0.144559 \\ 2.064686 \\ 41.765478 \\ 146.448441 \\ 249.705026 \end{bmatrix} \qquad a = 0.01$$

Kontynuacja na następnej stronie....

Przypadek osobliwości macierzy bezwładności
Ten przypadek ma miejsce gdy pominiemy bezwładność obrotową w macierzy mas

$M:=Mr$ *- przyjmujemy macierz diagonalną, bez bezwładności obrotowej*

Uwzględnienie warunków brzegowych

$i:=1..Lr$

$w:=1$ $\quad K_{w,i}:=0 \quad K_{i,w}:=0 \quad K_{w,w}:=1 \quad M_{w,i}:=0 \quad M_{w,w}:=1$

$w:=Lr-1$ $\quad K_{w,i}:=0 \quad K_{i,w}:=0 \quad K_{w,w}:=1 \quad M_{w,i}:=0 \quad M_{w,w}:=1$

$\kappa\cdot\left[K-\omega^2\ \beta_r\cdot M\right]\ y=0$ $\qquad |M|=0$ $\qquad |K|=2.48832\cdot10^7$ $\qquad K1:=K^{-1}$

Ponieważ macierz bezwładności jest osobliwa, nie można doprowadzić poprzednim sposobem do standardowego zadania wartości własnych. Odwracamy w takim wypadku macierz sztywności i otrzymujemy standardową postać zadania:

$|K1\cdot K-\omega^2\cdot\beta_r\cdot K1\cdot M|=0$ ----------> $\sigma=\dfrac{1}{\omega^2\cdot\beta_d}$ $\qquad |K1\cdot M-\sigma\cdot I|=0$

$\sigma:=\text{eigenvals}(K1\cdot M)$ $\qquad \sigma:=\text{sort}(\sigma)$

Ponieważ dostaniemy teraz dużo zerowych wartości własnych to nie można ich użyć do obliczenia częstości własnych bo będziemy mieli dzielenie przez zero ---------------> $\omega=\sqrt{\dfrac{1}{\sigma\cdot\beta_d}}$

Trzeba odrzucić n0 zerowych wartości z początku wektora σ.
Aby to zrobić musimy obliczyć n0, albo lepiej n1, czyli pierwszy niezerowy wyraz w wektorze σ.
Można to zrobić "ręcznie" ale w przypadku większych zadań to jest kłopotliwe i zaburza automatyzm algorytmu.
(! Żadnych ręcznych "robótek" w komputerowych rachunkach!)

```
Find_n1(v) := | n1 ← 0
              | for i ∈ rows(v) .. 1
              |   | if v_i > 0
              |   |   | n1 ← i
              | n1
```

$n1:=Find\_n1(\sigma)$ $\qquad n1=7$

$\sigma1:=\text{submatrix}(\sigma,n1,\text{rows}(\sigma),1,1)$

Lub "elegancko": $\sigma1:=\text{submatrix}(\sigma,Find\_n1(\sigma),\text{rows}(\sigma),1,1)$

$$\omega := \mathrm{sort}\left(\sqrt{\frac{1}{\sigma 1 \cdot \beta_r}}\right) \qquad \omega = \begin{bmatrix} 11.630059 \\ 41.666667 \\ 41.666667 \\ 46.410707 \\ 102.740115 \\ 169.182036 \end{bmatrix} \frac{\mathbf{rad}}{\mathbf{s}} \qquad \omega_A = \begin{bmatrix} 11.63144 \\ 46.525761 \\ 104.682963 \\ 186.103045 \\ 290.786008 \\ 418.731852 \end{bmatrix} \frac{\mathbf{rad}}{\mathbf{s}}$$

Różnice wynikające z różnych metod obliczeniowych

$$i := 1 \,..\, \mathrm{rows}(\omega_A) \qquad \delta_i := \omega_{A_i} - \omega_i \qquad \delta = \begin{bmatrix} 1.381171 \cdot 10^{-3} \\ 4.859095 \\ 6.30163 \cdot 10 \\ 1.396923 \cdot 10^{2} \\ 1.880459 \cdot 10^{2} \\ 2.495498 \cdot 10^{2} \end{bmatrix} \frac{\mathbf{rad}}{\mathbf{s}}$$

$$N = 5 \qquad \omega_{A_1} - \omega_1 = \left(1.381171 \cdot 10^{-3}\right) \frac{\mathbf{rad}}{\mathbf{s}}$$

Różnice zmniejszają się wraz ze wzrostem liczby elementów skończonych dla N=40 mamy:

$$\delta_{40} = \begin{bmatrix} 0 \\ 0.00002 \\ 0.000227 \\ 0.001289 \\ 0.004981 \\ 0.015112 \end{bmatrix} \cdot \frac{\mathbf{rad}}{\mathbf{s}}$$

W skali logarytmicznej ta zbieżność jest"prawie liniowa"!

$$N := \begin{bmatrix} 4 \\ 8 \\ 16 \\ 20 \\ 32 \\ 40 \\ 64 \end{bmatrix} \qquad \delta := \begin{bmatrix} 3.548559 \cdot 10^{-3} \\ 1.992366 \cdot 10^{-4} \\ 1.211635 \cdot 10^{-5} \\ 4.946518 \cdot 10^{-6} \\ 7.520953 \cdot 10^{-7} \\ 3.077947 \cdot 10^{-7} \\ 4.685426 \cdot 10^{-8} \end{bmatrix}$$

za



Uwagi końcowe

Jak widać dla belek najwygodniej jest stosować macierz diagonalną ze "zmniejszonymi" bezwładnościami obrotowymi, oczywiście jeżeli chcemy uzyskać zgodność z wynikami analitycznymi dla belki Bernoulliego.
Przy macierzy konsekwentnej doprowadzenie do postaci standardowej jest możliwe przy zastosowaniu faktoryzacji Cholesky'ego, ale dla belek nie ma to chyba sensu.
Przy elementach powierzchniowych jest konieczne.